3.7: Complex Combination

3.7.1 Introduction

Complex combination is a useful technique for solving problems that have periodic boundary conditions or forcing functions. Periodic boundary conditions or disturbances are often encountered in engineering problems; for example, the response of the ground over the course of a day or an engine cylinder during a cycle. This type of problem was encountered in EXAMPLE 3.1-2; a temperature sensor (treated as a lumped capacitance) was exposed to an oscillating fluid temperature. In EXAMPLE 3.1-2, the problem is solved analytically and the temperature response of the sensor is found to be the sum of a homogeneous solution that decayes to zero (as time becomes sufficiently greater than the time constant of the sensor), and a particular solution that is the sustained response. Figure 3-36 illustrates the sensor response and shows that eventually only the particular (or sustained) solution persists. Complex combination is a convenient method for obtaining only this sustained solution, which is often the only portion of the solution that is of interest.



Figure 3-36: **Temperature of the sensor and surrounding fluid obtained in the solution to EXAMPLE 3.1-2.**

Complex combination can be used for transient problems that are 0-D (i.e., lumped), 1-D, and even 2-D or 3-D. A complete description of complex combination can be found in Myers (1998); a summary of this technique is presented in this section.

3.7.2 Complex-Variable Theory

It is necessary to review some key formulae from complex-variable theory in order to accomplish complex combination. However, the use of Maple and EES allows complex combination to be accomplished without requiring much complex algebra.

A complex number can converted to exponential form by using the mathematical identities:

$$\exp(i\,x) = \cos(x) + i\sin(x) \tag{3-422}$$

$$\exp(-i\,x) = \cos(x) - i\sin(x) \tag{3-423}$$

Algebraic manipulation of Eqs. (3-422) and (3-423) allow the functions sine and cosine to be written as:

$$\cos\left(x\right)=\frac{1}{2}\exp\left(i\,x\right)+\frac{1}{2}\exp\left(-i\,x\right) \tag{3-424}$$

$$\sin\left(x\right)=-\frac{1}{2}i\exp\left(i\,x\right)+\frac{1}{2}i\exp\left(-i\,x\right) \tag{3-425}$$

where i is the square root of negative one, $\sqrt{-1}$.

3.7.3 Complex Combination

In order to use the method of complex combination, it is necessary to first derive the differential equations and boundary conditions that govern the problem. Because only the sustained solution is of interest, it is not necessary to specify an initial condition. The problem must be completely homogeneous except for the periodic component. That is, if the periodic term is removed, then both the governing differential equation and all of the boundary conditions must be homogeneous. In many cases, a problem can be made homogeneous by using a simple transformation that is obvious from inspection (as discussed in Section 2.2.2) or by carefully dividing the problem statement into its non-homogeneous and homogeneous parts (as discussed in Sections 2.3.2 and 3.5.5).

The solution T_{sus} is the sustained, periodic solution to the problem with a single non-homogeneous term that is periodic with the form A sin($\omega\, t$) or A cos($\omega\, t$). The steps associated with obtaining T_{sus} with the method of complex combination are summarized below and then applied to 0-D and 1-D transient problems.

1. Construct a problem for $T_{90°}$ that is identical to the problem for T_{sus} but with a periodic component that is 90° out of phase. That is, if the original disturbance is $A\sin\left(\omega t\right)$ then it is replaced with $A\cos\left(\omega t\right)$ and if the original disturbance is $A\cos\left(\omega t\right)$ then it is replaced with $A\sin\left(\omega t\right)$.
2. Construct the complex conjugate problem for T_{cc}, where:

$$T_{cc}=T+i\,T_{90°} \tag{3-426}$$

   The complex conjugate problem is obtained by multiplying the problem for $T_{90°}$ by i ($\sqrt{-1}$) and adding it to the problem for T_{sus} (the governing differential equation and boundary conditions must both be constructed in this manner).
3. Convert the complex conjugate problem to exponential form.
4. Assume an exponential solution for T_{cc}. This will result in the time components of the problem canceling out and therefore will lead to an algebraic problem (for a 0-D problem) or an ordinary differential equation (for a 1-D problem).
5. The sustained solution is the real part of the solution.

These steps become clearer when illustrated by example.

Complex Combination for 0-D Problems

A temperature sensor exposed to an oscillating fluid temperature is studied in EXAMPLE 3.1-2. In this section, the sustained response of the temperature sensor is obtained using the method of complex combination. The problem statement is repeated here for convenience. The temperature of the fluid varies in a sinusoidal manner with mean temperature $\overline{T}_\infty$=320°C, amplitude ΔT_∞ =50°C, and frequency f = 0.5 Hz (ω = 3.14 rad/s). The sensor can be modeled as a sphere of diameter D = 1.0 mm. The sensor is made of a material with conductivity k_s = 50 W/m-K, specific heat capacity c_s = 150 J/kg-K, and density ρ_s = 16000 kg/m^3. In order to provide corrosion resistance, the sensor has been coated with a thin layer of plastic. The coating is th_c = 100 μm thick with conductivity k_c = 0.2 W/m-K and has negligible heat capacity relative to the sensor itself. The heat transfer coefficient between the surface of the coating and the fluid is $\overline{h}$ = 500 W/m^2-K.

The inputs are entered in EES:

```
$UnitSystem SI MASS RAD PA  K J
$TABSTOPS   0.2 0.4 0.6 0.8 3.5 in

"Inputs"
T_infinity_bar=converttemp(C,K,320 [C])          "average temperature of the fluid"
DELTAT_infinity=50 [K]                           "amplitude of fluid temperature change"
f=0.5 [cycle/s]                                  "frequency of fluid temperature change"
D=1.0 [mm]*convert(mm,m)                         "diameter of sensor"
k_s=50 [W/m-K]                                   "conductivity of sensor material"
c_s=150 [J/kg-K]                                 "specific heat capacity of sensor material"
rho_s=16000 [kg/m^3]                             "density of sensor material"
th_c=100 [micron]*convert(micron,m)              "thickness of coating"
k_c=0.2 [W/m-K]                                  "conductivity of coating"
h_bar=500 [W/m^2-K]                              "heat transfer coefficient"
omega=f*convert(cycle/s,rad/s)                   "angular frequency"
```

In EXAMPLE 3.1-2, it was shown that a lumped capacitance model of the sensor was appropriate because the Biot number is much less than 1.0 and therefore this is a 0-D, transient problem. The time constant of the sensor was given by:

$$\tau_{lumped} = \left(R_{cond,c} + R_{conv}\right) C \tag{3-427}$$

where $R_{cond,c}$ is the conduction resistance of the coating:

$$R_{cond,c} = \frac{\left[\frac{2}{D} - \frac{2}{\left(D + 2\,th_c\right)}\right]}{4\,\pi\,k_c} \tag{3-428}$$

R_{conv} is the resistance to heat transfer by convection from the surface of the coating:

$$R_{conv} = \frac{1}{\bar{h}\,4\,\pi\left(\frac{D}{2}+th_c\right)^2} \tag{3-429}$$

and C is the total heat capacity of the sensor:

$$C = \frac{4\,\pi}{3}\left(\frac{D}{2}\right)^3 \rho_s\, c_s \tag{3-430}$$

These preliminary calculations are entered in EES:

```
R_conv=1/(h_bar*4*pi*(D/2+th_c)^2)                 "convective resistance"
R_cond_c=(1/(D/2)-1/(D/2+th_c))/(4*pi*k_c)         "conduction resistance of coating"
C=4*pi*(D/2)^3*rho_s*c_s/3                         "capacitance of the sensor"
tau=(R_conv+R_cond_c)*C                            "time constant of the sensor"
```

The governing differential equation for the sensor temperature (from EXAMPLE 3.1-2) is:

$$\frac{dT}{dt}+\frac{T}{\tau_{lumped}}=\frac{\bar{T}_\infty}{\tau_{lumped}}+\frac{\Delta T_\infty \sin(\omega\, t)}{\tau_{lumped}} \tag{3-431}$$

The method of complex combination requires that the problem be homogeneous if the periodic disturbance is removed. If the sinusoidal term is removed from Eq. (3-431) then the ordinary differential equation becomes:

$$\frac{dT}{dt}+\frac{T}{\tau_{lumped}}=\frac{\bar{T}_\infty}{\tau_{lumped}} \tag{3-432}$$

which is not homogeneous. However, it is possible to transform Eq. (3-432) by defining the temperature difference relative to the mean fluid temperature:

$$\theta = T-\bar{T}_\infty \tag{3-433}$$

Substituting Eq. (3-433) into Eq. (3-431) leads to:

$$\frac{d\theta}{dt}+\frac{\theta}{\tau_{lumped}}=\frac{\Delta T_\infty}{\tau_{lumped}}\sin(\omega\, t) \tag{3-434}$$

Equation (3-434) is homogeneous if the sine term is removed.

The first step in the application of complex combination is to construct a problem for θ_{90° by replacing the sine term in Eq. (3-434) with a cosine term:

$$\frac{d\theta_{90^\circ}}{dt}+\frac{\theta_{90^\circ}}{\tau_{lumped}}=\frac{\Delta T_\infty}{\tau_{lumped}}\cos\left(\omega t\right) \tag{3-435}$$

The complex conjugate problem for θ_{cc} is constructed from θ and θ_{90° according to:

$$\theta_{cc}=\theta+i\,\theta_{90^\circ} \tag{3-436}$$

The problem statement for the complex conjugate problem is obtained by multiplying the governing equation for θ_{90°, Eq. (3-435) by i (the $\sqrt{-1}$) and adding it to the governing equation for θ, Eq. (3-434):

$$\frac{d\theta}{dt}+\frac{\theta}{\tau_{lumped}}+i\left(\frac{d\theta_{90^\circ}}{dt}+\frac{\theta_{90^\circ}}{\tau_{lumped}}\right)=\frac{\Delta T_\infty}{\tau_{lumped}}\sin\left(\omega t\right)+i\left[\frac{\Delta T_\infty}{\tau_{lumped}}\cos\left(\omega t\right)\right] \tag{3-437}$$

or

$$\frac{d\overbrace{\left(\theta+i\,\theta_{90^\circ}\right)}^{\theta_{cc}}}{dt}+\frac{\overbrace{\left(\theta+i\,\theta_{90^\circ}\right)}^{\theta_{cc}}}{\tau_{lumped}}=\frac{\Delta T_\infty}{\tau_{lumped}}\left[\sin\left(\omega t\right)+i\cos\left(\omega t\right)\right] \tag{3-438}$$

Substituting Eq. (3-436) into Eq. (3-438) leads to the complex conjugate problem:

$$\frac{d\theta_{cc}}{dt}+\frac{\theta_{cc}}{\tau_{lumped}}=\frac{\Delta T_\infty}{\tau_{lumped}}\left[\sin\left(\omega t\right)+i\cos\left(\omega t\right)\right] \tag{3-439}$$

Equation (3-439) is in complex form; the next step of the process is to convert the complex conjugate problem to exponential form. Substituting Eqs. (3-424) and (3-425) into Eq. (3-439) leads to:

$$\frac{d\theta_{cc}}{dt}+\frac{\theta_{cc}}{\tau_{lumped}}=\frac{\Delta T_\infty}{\tau_{lumped}}\left[\underbrace{-\frac{1}{2}i\exp\left(i\,\omega t\right)+\frac{1}{2}i\exp\left(-i\,\omega t\right)}_{\sin(\omega t)}+\underbrace{i\frac{1}{2}\exp\left(i\,\omega t\right)+i\frac{1}{2}\exp\left(-i\,\omega t\right)}_{\cos(\omega t)}\right] \tag{3-440}$$

or

$$\frac{d\theta_{cc}}{dt}+\frac{\theta_{cc}}{\tau_{lumped}}=\frac{\Delta T_\infty}{\tau_{lumped}}i\exp\left(-i\,\omega t\right) \tag{3-441}$$

which is the exponential form of the complex conjugate problem. Maple can be used to accomplish this conversion process. The original form of the complex conjugate problem, Eq. (3-439), is entered in Maple (note that $\sqrt{-1}$ is indicated by a capital I in Maple).

```
> restart;
> ODEcc:=diff(theta_cc(t),t)+theta_cc(t)/tau=DELTAT_infinity*(sin(omega*t)+I*cos(omega*t))/tau;
```

$$ODEcc := \left(\frac{d}{dt}\,\text{theta\_cc}(t)\right) + \frac{\text{theta\_cc}(t)}{\tau} = \frac{DELTAT\_infinity\,(\sin(\omega\, t) + \cos(\omega\, t)\, I)}{\tau}$$

The problem can be converted to exponential form using the convert command in Maple with the exp identifier:

```
> ODEcc:=convert(ODEcc,exp);
```

$$ODEcc := \left(\frac{d}{dt}\,\text{theta\_cc}(t)\right) + \frac{\text{theta\_cc}(t)}{\tau} =$$

$$\frac{DELTAT\_infinity\left(-\frac{1}{2}\, I\left(\mathbf{e}^{(\omega\, t\, I)} - \frac{1}{\mathbf{e}^{(\omega\, t\, I)}}\right) + \left(\frac{1}{2}\,\mathbf{e}^{(\omega\, t\, I)} + \frac{1}{2}\,\frac{1}{\mathbf{e}^{(\omega\, t\, I)}}\right) I\right)}{\tau}$$

The result does not appear to match Eq. (3-441); however, the simplify command shows that the two results are actually equivalent:

```
> ODEcc:=simplify(ODEcc);
```

$$ODEcc := \frac{\left(\frac{d}{dt}\,\text{theta\_cc}(t)\right)\tau + \text{theta\_cc}(t)}{\tau} = \frac{DELTAT\_infinity\,\mathbf{e}^{(-I\,\omega\, t)}\, I}{\tau}$$

An exponential solution is assumed for θ_{cc}. The form of the solution should be selected so that the time term cancels out of the complex conjugate problem. If the exponential form of the problem contains a term with $\exp(i\,\omega t)$ then $\theta_{cc} = B\exp(i\,\omega t)$ should be assumed. An exponential problem with $\exp(-i\,\omega t)$ would lead to $\theta_{cc} = B\exp(-i\,\omega t)$. Examining Eq. (3-441) shows that the form of the solution should be:

$$\theta_{cc} = B\exp(-i\,\omega\, t) \tag{3-442}$$

The derivative of Eq. (3-442) with respect to time is:

$$\frac{d\theta_{cc}}{dt} = -B\, i\,\omega\exp(-i\,\omega\, t) \tag{3-443}$$

Substituting Eqs. (3-442) and (3-443) into Eq. (3-441) leads to:

$$-B\,i\,\omega\exp\left(-i\,\omega\,t\right)+\frac{B}{\tau_{lumped}}\exp\left(-i\,\omega\,t\right)=\frac{\Delta T_{\infty}}{\tau_{lumped}}i\exp\left(-i\,\omega\,t\right) \quad (3\text{-}444)$$

Equation (3-444) can be simplified to an algebraic equation for B by dividing through by $\exp\left(-i\,\omega\,t\right)$:

$$-B\,i\,\omega+\frac{B}{\tau_{lumped}}=\frac{\Delta T_{\infty}}{\tau_{lumped}}i \quad (3\text{-}445)$$

Solving Eq. (3-445) for B leads to:

$$B=\frac{\dfrac{\Delta T_{\infty}}{\tau_{lumped}}i}{\left(\dfrac{1}{\tau_{lumped}}-i\,\omega\right)} \quad (3\text{-}446)$$

Substituting Eq. (3-446) into Eq. (3-442) leads to the solution to the complex conjugate problem:

$$\theta_{cc}=\frac{\dfrac{\Delta T_{\infty}}{\tau_{lumped}}i}{\left(\dfrac{1}{\tau_{lumped}}-i\,\omega\right)}\exp\left(-i\,\omega\,t\right) \quad (3\text{-}447)$$

The sustained solution to the original problem is the real part of the complex conjugate solution. Some complex algebra is required to explicitly identify the real portion of the solution by hand. Substituting Eq. (3-423) into Eq. (3-447) leads to:

$$\theta_{cc}=\frac{\dfrac{\Delta T_{\infty}}{\tau_{lumped}}i}{\left(\dfrac{1}{\tau_{lumped}}-i\,\omega\right)}\left[\cos\left(\omega\,t\right)-i\sin\left(\omega\,t\right)\right] \quad (3\text{-}448)$$

Multiplying the numerator and denominator of Eq. (3-448) by the complex conjugate of the denominator will make the denominator real:

$$\theta_{cc}=\frac{\dfrac{\Delta T_{\infty}}{\tau_{lumped}}i\left(\dfrac{1}{\tau_{lumped}}+i\,\omega\right)}{\left(\dfrac{1}{\tau_{lumped}}-i\,\omega\right)\left(\dfrac{1}{\tau_{lumped}}+i\,\omega\right)}\left[\cos\left(\omega\,t\right)-i\sin\left(\omega\,t\right)\right] \quad (3\text{-}449)$$

which can be simplified to:

$$\theta_{cc}=\frac{\frac{\Delta T_{\infty}}{\tau_{lumped}}i\left[\frac{\cos(\omega t)}{\tau_{lumped}}-\frac{i\sin(\omega t)}{\tau_{lumped}}+i\,\omega\cos(\omega t)-i^{2}\,\omega\sin(\omega t)\right]}{\left(\frac{1}{\tau_{lumped}^{2}}+\omega^{2}\right)} \tag{3-450}$$

Equation (3-450) can be broken into its complex and real parts:

$$\theta_{cc}=\frac{\Delta T_{\infty}\left[\frac{\sin(\omega t)}{\tau_{lumped}^{2}}-\frac{\omega}{\tau_{lumped}}\cos(\omega t)\right]}{\left(\frac{1}{\tau_{lumped}^{2}}+\omega^{2}\right)}+\frac{\Delta T_{\infty}\left[\frac{\cos(\omega t)}{\tau_{lumped}^{2}}+\frac{\omega}{\tau_{lumped}}\sin(\omega t)\right]}{\left(\frac{1}{\tau_{lumped}^{2}}+\omega^{2}\right)}i \tag{3-451}$$

According to Eq. (3-436), the real part of Eq. (3-451) is the sustained solution for θ:

$$\theta_{sus}=\frac{\Delta T_{\infty}\left[\frac{\sin(\omega t)}{\tau_{lumped}^{2}}-\frac{\omega}{\tau_{lumped}}\cos(\omega t)\right]}{\left(\frac{1}{\tau_{lumped}^{2}}+\omega^{2}\right)} \tag{3-452}$$

The complex algebra can be accomplished or checked using Maple; Eq. (3-447) is entered into Maple:

```
> restart;
> theta_cc:=(DELTAT_infinity*I/tau/(1/tau-I*omega))*exp(-I*omega*t);
```

$$theta\_cc := \frac{DELTAT\_infinity\ \mathrm{e}^{(-I\,\omega\,t)}\ I}{\tau\left(\frac{1}{\tau}-\omega\ I\right)}$$

and the evalc command is used to identify the real and imaginary parts:

```
> evalc(theta_cc);
```

$$-\frac{DELTAT\_infinity\ \omega\ \cos(\omega\ t)}{\tau\left(\frac{1}{\tau^{2}}+\omega^{2}\right)}+\frac{DELTAT\_infinity\ \sin(\omega\ t)}{\tau^{2}\left(\frac{1}{\tau^{2}}+\omega^{2}\right)}+\left(\frac{DELTAT\_infinity\ \cos(\omega\ t)}{\tau^{2}\left(\frac{1}{\tau^{2}}+\omega^{2}\right)}+\frac{DELTAT\_infinity\ \omega\ \sin(\omega\ t)}{\tau\left(\frac{1}{\tau^{2}}+\omega^{2}\right)}\right)I$$

The real term can be copied from Maple and pasted into EES for evaluation:

```
t=0 [s]                                                              "time"
theta_sus=-DELTAT_infinity/tau*omega/(1/(tau^2)+omega^2)*cos(omega*t)&
    +DELTAT_infinity/tau^2/(1/(tau^2)+omega^2)*sin(omega*t)
    "sustained solution to homogeneous equation, from Maple"
T_sus=T_infinity_bar+theta_sus                                       "sustained solution"
T_sus_C=converttemp(K,C,T_sus)                                       "in C"
T_infinity=T_infinity_bar+DELTAT_infinity*sin(omega*t)               "fluid temperature"
T_infinity_C=converttemp(K,C,T_infinity)                             "in C"
```

A parametric table is setup that contains the variables t, T_sus_C, and T_f_C. The sustained solution and fluid temperature are shown in Figure 3-37 as a function of time. Notice that the sustained solution is identical to the particular solution obtained EXAMPLE 3.1-2.



Figure 3-37: The sustained solution for the response of the sensor to an oscillating fluid temperature.

Maple is useful for automatically converting between complex forms of the problem and the complex conjugate solution. EES continues to be the most convenient method for numerically evaluating the symbolic solution for a particular set of parameters.

Complex Combination for 1-D Problems

The method of complex combination can also be applied to 1-D transient problems in order to obtain the sustained response to a periodic disturbance. The steps remain the same as those used for 0-D problems. In this section, the technique will be illustrated in the context of the problem illustrated in Figure 3-38. The upper surface of a piston (at x=0) is exposed to a sinusoidally varying gas temperature according to:

$$T_c = \overline{T}_c + \Delta T_c \sin\left(\omega t\right) \tag{3-453}$$

where $\overline{T}_c$ = 800 K is the average temperature of the gas in the cylinder and ΔT_c = 500 K is the amplitude of the temperature variation. The gas temperature fluctuates with an angular velocity of ω = 30 rad/s. The average heat transfer coefficient between the gas in the cylinder and the upper surface of the piston is $\overline{h}_c$ = 5800 W/m^2-K. The lower surface of piston (at x = L) is cooled by fluid at T_o = 375 K. The average heat transfer coefficient between the lower surface of the piston and the fluid is $\overline{h}_o$ = 350 W/m^2-K. The piston material has density ρ = 2800 kg/m^3, conductivity k = 67 W/m-K, and specific heat capacity c = 340 J/kg-K. The thickness of the piston in the x-direction is L = 2.0 cm.



Figure 3-38: Piston subjected to a sinusoidal gas temperature.

The inputs are entered in EES:

```
$UnitSystem SI MASS RAD PA  K J
$TABSTOPS   0.2 0.4 0.6 0.8 3.5 in
$COMPLEX On i

"Inputs"
L=2.0 [cm]*convert(cm,m)                "cylinder thickness"
k=67 [W/m-K]                            "thermal conductivity"
rho=2800 [kg/m^3]                       "density"
c=340 [J/kg-K]                          "specific heat capacity"
omega=30 [rad/s]                        "angular frequency of the fluid temperature fluctuation"
T_c_bar=800 [K]                         "average cylinder temperature"
DT_c=500 [K]                            "amplitude of the cylinder temperature variation"
h_bar_c=5800 [W/m^2-K]                  "heat transfer coefficient in the cylinder"
T_o=375 [K]                             "back-side air temperature"
h_bar_o=350 [W/m^2-K]                   "heat transfer coefficient on the cooled side"
alpha=k/(rho*c)                         "thermal diffusivity"
```

This problem will take advantage of EES' complex algebra feature. To activate the complex algebra capability in EES, select Preferences from the Options menu and use the arrows to reach the Complex Tab. Check the Do complex algebra box and select the variable i to be the imaginary number. Complex algebra can also be activated by using the $COMPLEX On directive, optionally followed by i or j, as shown in the EES code above.

Select Solve from the Calculate menu and then select Variable Info from the Options menu. Notice that each parameter that has been entered now corresponds to two variables in EES (for example, the variable k has become k_i and k_r). The variable subscripted with i corresponds to its imaginary part and the variable subscripted r corresponds to its real part. Because there was no imaginary component associated with any of the physical quantities used in the problem specification, all of the variables subscripted with i are zero (e.g., k_i = 0). EES' ability to do complex algebra simplifies the process of finding the real part of the complex conjugate solution.

The governing differential equation is obtained using an energy balance on the differential control volume shown in Figure 3-38:

$$\dot{q}_x = \dot{q}_{x+dx} + \frac{\partial U}{\partial t} \tag{3-454}$$

Expanding the $x+dx$ and substituting the appropriate rate equations into Eq. (3-454) leads to:

$$0 = \frac{\partial}{\partial x}\left(-k\,A\frac{\partial T}{\partial x}\right)dx + \frac{\partial}{\partial t}\left(A\,dx\,\rho\,c\,T\right) \tag{3-455}$$

or

$$\alpha\frac{\partial^2 T}{\partial x^2} = \frac{\partial T}{\partial t} \tag{3-456}$$

The spatial boundary conditions are obtained with interface balances at the upper (x = 0) and lower (x = L) surfaces of the piston:

$$\overline{h}_c\left(\overline{T}_c + \Delta T_c \sin\left(\omega\,t\right) - T_{x=0}\right) = -k\left.\frac{\partial T}{\partial x}\right|_{x=0} \tag{3-457}$$

$$-k\left.\frac{\partial T}{\partial x}\right|_{x=L} = \overline{h}_o\left(T_{x=L} - T_o\right) \tag{3-458}$$

No initial condition is required since only the sustained, periodic solution will be obtained. The method of complex combination can only be applied if the problem is homogeneous except for the periodic disturbance. Neither of the boundary conditions, Eqs. (3-457) and (3-458), are homogeneous and there is no obvious transformation that will eliminate the non-homogeneous terms. Therefore, it is necessary to carefully decompose the solution into a homogeneous portion, $T_h(x,t)$, that can be solved using complex combination and a non-homogeneous portion, $X(x)$, that is solved separately. This process is discussed in Section 3.5.5:

$$T\left(x,t\right) = T_h\left(x,t\right) + X\left(x\right) \tag{3-459}$$

Substituting Eq. (3-459) into the governing differential equation, Eq. (3-456), leads to:

$$\underbrace{\alpha\frac{d^2 X}{dx^2}}_{\text{ODE for } X} + \underbrace{\alpha\frac{\partial^2 T_h}{\partial x^2} = \frac{\partial T_h}{\partial t}}_{\text{homogeneous PDE}} \tag{3-460}$$

The governing differential equation for T_h must remain homogeneous; therefore:

$$\alpha \frac{\partial^2 T_h}{\partial x^2} = \frac{\partial T_h}{\partial t} \tag{3-461}$$

which leaves the governing differential equation for the particular portion of the problem:

$$\frac{d^2 X}{dx^2} = 0 \tag{3-462}$$

The general solution for X is obtained by integrating twice:

$$X = C_2 + C_1 x \tag{3-463}$$

where C_1 and C_2 are constants of integration. Equation (3-459) is substituted into the boundary condition at $x = 0$, Eq. (3-457):

$$\bar{h}_c \left[\bar{T}_c + \Delta T_c \sin(\omega t) - T_{h,x=0} - X_{x=0} \right] = -k \left. \frac{\partial T_h}{\partial x} \right|_{x=0} - k \left. \frac{dX}{dx} \right|_{x=0} \tag{3-464}$$

The boundary condition for T_h must be homogeneous except for the periodic term:

$$\bar{h}_c \left[\Delta T_c \sin(\omega t) - T_{h,x=0} \right] = -k \left. \frac{\partial T_h}{\partial x} \right|_{x=0} \tag{3-465}$$

Subtracting Eq. (3-465) from Eq. (3-464) leads to the first boundary condition for X:

$$\bar{h}_c \left(\bar{T}_c - X_{x=0} \right) = -k \left. \frac{dX}{dx} \right|_{x=0} \tag{3-466}$$

Substituting Eq. (3-459) into the boundary condition at $x = L$, Eq. (3-458), leads to:

$$-k \left. \frac{\partial T_h}{\partial x} \right|_{x=L} - k \left. \frac{dX}{dx} \right|_{x=L} = \bar{h}_o \left(T_{h,x=L} + X_{x=L} - T_o \right) \tag{3-467}$$

The boundary condition for T_h must be homogeneous:

$$-k \left. \frac{\partial T_h}{\partial x} \right|_{x=L} = \bar{h}_o T_{h,x=L} \tag{3-468}$$

Subtracting Eq. (3-468) from Eq. (3-467) leads to the second boundary condition for X:

$$-k \left. \frac{dX}{dx} \right|_{x=L} = \bar{h}_o \left(X_{x=L} - T_o \right) \tag{3-469}$$

Substituting the general solution for X, Eq. (3-463), into the boundary conditions for X, Eqs. (3-466) and (3-469), leads to two equations for the two, unknown constants of integration, C_1 and C_2:

$$\bar{h}_c\left(\bar{T}_c - C_2\right) = -k\,C_1 \tag{3-470}$$

$$-k\,C_1 = \bar{h}_o\left(C_1\,L + C_2 - T_o\right) \tag{3-471}$$

Equations (3-470) and (3-471) are entered in EES in order to determine the constants C_1 and C_2 that define the function X.

```
"determine constants for particular portion of the solution"
h_bar_c*(T_c_bar-C_2)=-k*C_1                              "BC for X at x=0"
-k*C_1=h_bar_o*(C_1*L+C_2-T_o)                            "BC for X at x=L"
```

The solution for the homogeneous portion of the solution is obtained using the method of complex combination. The problem for $T_{h,90^\circ}$ is constructed by replacing the periodic component of the T_h problem, Eqs. (3-461), (3-465), and (3-468), with one that is 90° out of phase (i.e., by replacing the $\sin\left(\omega t\right)$ term in the boundary condition at $x = 0$ with a $\cos\left(\omega t\right)$ term):

$$\alpha\,\frac{\partial^2 T_{h,90^\circ}}{\partial x^2} = \frac{\partial T_{h,90^\circ}}{\partial t} \tag{3-472}$$

$$\bar{h}_c\left[\Delta T_c\cos\left(\omega t\right) - T_{h,90^\circ,x=0}\right] = -k\left.\frac{\partial T_{h,90^\circ}}{\partial x}\right|_{x=0} \tag{3-473}$$

$$-k\left.\frac{\partial T_{h,90^\circ}}{\partial x}\right|_{x=L} = \bar{h}_o\,T_{h,90^\circ,x=L} \tag{3-474}$$

The complex conjugate problem for $T_{h,cc}$:

$$T_{h,cc} = T_h + i\,T_{h,90^\circ} \tag{3-475}$$

is constructed by multiplying the governing differential equation and boundary conditions for $T_{h,90^\circ}$, Eqs. (3-472) through (3-474), by i (the square root of negative one) and adding it to the governing differential equation and boundary conditions for T_h, Eqs. (3-461), (3-465), and (3-468):

$$\alpha \frac{\partial^2 T_h}{\partial x^2} + i\,\alpha \frac{\partial^2 T_{h,90^\circ}}{\partial x^2} = \frac{\partial T_h}{\partial t} + i \frac{\partial T_{h,90^\circ}}{\partial t} \tag{3-476}$$

$$\bar{h}_c \left[\Delta T_c \sin(\omega t) - T_{h,x=0} \right] + i\,\bar{h}_c \left[\Delta T_c \cos(\omega t) - T_{h,90^\circ,x=0} \right] = -k \left.\frac{\partial T_h}{\partial x}\right|_{x=0} - i\,k \left.\frac{\partial T_{h,90^\circ}}{\partial x}\right|_{x=0} \tag{3-477}$$

$$-k \left.\frac{\partial T_h}{\partial x}\right|_{x=L} - i\,k \left.\frac{\partial T_{h,90^\circ}}{\partial x}\right|_{x=L} = \bar{h}_o\, T_{h,x=L} + i\,\bar{h}_o\, T_{h,90^\circ,x=L} \tag{3-478}$$

Subsituting Eq. (3-475) into Eqs. (3-476) through (3-478) leads to the complex conjugate problem:

$$\alpha \frac{\partial^2 T_{h,cc}}{\partial x^2} = \frac{\partial T_{h,cc}}{\partial t} \tag{3-479}$$

$$\bar{h}_c \left[\Delta T_c \left(\sin(\omega t) + i \cos(\omega t) \right) - T_{h,cc,x=0} \right] = -k \left.\frac{\partial T_{h,cc}}{\partial x}\right|_{x=0} \tag{3-480}$$

$$-k \left.\frac{\partial T_{h,cc}}{\partial x}\right|_{x=L} = \bar{h}_o\, T_{h,cc,x=L} \tag{3-481}$$

The periodic variation in Eq. (3-480) must be converted to exponential form; this is most easily accomplished using Maple:

```
> restart;
> convert(sin(omega*t)+I*cos(omega*t),exp);
```

$$-\frac{1}{2} I \left(e^{(\omega\, t\, I)} - \frac{1}{e^{(\omega\, t\, I)}} \right) + \left(\frac{1}{2} e^{(\omega\, t\, I)} + \frac{1}{2} \frac{1}{e^{(\omega\, t\, I)}} \right) I$$

```
> simplify(%);
```

$$e^{(-I\, \omega\, t)}\, I$$

which leads to:

$$\bar{h}_c \left[i\, \Delta T_c \exp(-i\,\omega t) - T_{h,cc,x=0} \right] = -k \left.\frac{\partial T_{h,cc}}{\partial x}\right|_{x=0} \tag{3-482}$$

The complex conjugate solution is assumed in the appropriate exponential form:

$$T_{h,cc} = B(x)\exp(-i\,\omega\,t) \tag{3-483}$$

Notice that the coefficient *B* multiplying the time component must be a function of *x* because this is a 1-D problem. Substituting Eq. (3-483) into the partial differential equation, Eq. (3-479), leads to:

$$\alpha\,\frac{\partial^2}{\partial x^2}\left[B(x)\exp(-i\,\omega\,t)\right] = \frac{\partial}{\partial t}\left[B(x)\exp(-i\,\omega\,t)\right] \tag{3-484}$$

or

$$\alpha\exp(-i\,\omega\,t)\frac{d^2B}{dx^2} = -B\,i\,\omega\exp(-i\,\omega\,t) \tag{3-485}$$

Note that the time variation must cancel from the complex conjugate problem when expressed in this manner (a consequence of the problem being homogeneous), leaving an ordinary differential equation for *B*:

$$\alpha\,\frac{d^2B}{dx^2} = -B\,i\,\omega \tag{3-486}$$

Substituting Eq. (3-483) into the boundary condition at $x = L$, Eq. (3-481), leads to:

$$-k\,\frac{\partial}{\partial x}\left[B(x)\exp(-i\,\omega\,t)\right]_{x=L} = \overline{h}_o\left[B(x)\exp(-i\,\omega\,t)\right]_{x=L} \tag{3-487}$$

The time term cancels out of Eq. (3-487), leaving:

$$-k\left.\frac{dB}{dx}\right|_{x=L} = \overline{h}_o\,B_{x=L} \tag{3-488}$$

Substituting Eq. (3-483) into the boundary condition at $x = 0$, Eq. (3-482), leads to:

$$\overline{h}_c\left[i\,\Delta T_c\exp(-i\,\omega\,t) - B_{x=0}\exp(-i\,\omega\,t)\right] = -k\left.\frac{dB}{dx}\right|_{\mathrm{x=0}}\exp(-i\,\omega\,t) \tag{3-489}$$

again the time terms cancel, leaving:

$$\overline{h}_c\left(i\,\Delta T_c - B_{x=0}\right) = -k\left.\frac{dB}{dx}\right|_{\mathrm{x=0}} \tag{3-490}$$

Equations (3-488) and (3-490) provide boundary conditions for the ordinary differential equation for $B(x)$, Eq. (3-486); the problem for B can be solved using Maple. The ordinary differential equation is entered:

```
> restart;
> ODE:=alpha*diff(diff(B(x),x),x)=-omega*B(x)*I;
```

$$ODE := \alpha \left(\frac{d^2}{dx^2} \mathrm{B}(x) \right) = -I\,\omega\,\mathrm{B}(x)$$

and solved using the dsolve command:

```
> B_s:=dsolve(ODE);
```

$$B\_s := \mathrm{B}(x) = \_C1 \sin\left(\frac{\left(\frac{1}{2} + \frac{1}{2} I\right) \sqrt{2}\,\sqrt{\omega}\,x}{\sqrt{\alpha}} \right) + \_C2 \cos\left(\frac{\left(\frac{1}{2} + \frac{1}{2} I\right) \sqrt{2}\,\sqrt{\omega}\,x}{\sqrt{\alpha}} \right)$$

The two undetermined constants must be obtained by considering the boundary conditions. Equations (3-488) and (3-490) are evaluated symbolically in Maple:

```
> -k*rhs(eval(diff(B_s,x),x=L))=h_bar_o*rhs(eval(B_s,x=L));
```

$$-k \left(\frac{\left(\frac{1}{2} + \frac{1}{2} I\right) \_C1 \cos\left(\frac{\left(\frac{1}{2} + \frac{1}{2} I\right) \sqrt{2}\,\sqrt{\omega}\,L}{\sqrt{\alpha}} \right) \sqrt{2}\,\sqrt{\omega}}{\sqrt{\alpha}} - \frac{\left(\frac{1}{2} + \frac{1}{2} I\right) \_C2 \sin\left(\frac{\left(\frac{1}{2} + \frac{1}{2} I\right) \sqrt{2}\,\sqrt{\omega}\,L}{\sqrt{\alpha}} \right) \sqrt{2}\,\sqrt{\omega}}{\sqrt{\alpha}} \right) =$$
$$h\_bar\_o \left(\_C1 \sin\left(\frac{\left(\frac{1}{2} + \frac{1}{2} I\right) \sqrt{2}\,\sqrt{\omega}\,L}{\sqrt{\alpha}} \right) + \_C2 \cos\left(\frac{\left(\frac{1}{2} + \frac{1}{2} I\right) \sqrt{2}\,\sqrt{\omega}\,L}{\sqrt{\alpha}} \right) \right)$$

```
> h_bar_c*(DT_c*I-rhs(eval(B_s,x=0)))=-k*rhs(eval(diff(B_s,x),x=0));
```

$$h\_bar\_c\,(DT\_c\,I - \_C2) = \frac{\left(\frac{-1}{2} - \frac{1}{2} I\right) k\,\_C1\,\sqrt{2}\,\sqrt{\omega}}{\sqrt{\alpha}}$$

The boundary conditions are copied and pasted into EES in order to numerically evaluate the constants; note that the variables _C1 and _C2 from Maple must be changed to C_3 and C_4 in EES (the solution for X, Eq. (3-463), already includes constant C_1 and C_2).

```
"boundary condition #1 from Maple"
-k*((1/2+1/2*I)*C_3*cos((1/2+1/2*I)*2^(1/2)*omega^(1/2)/alpha^(1/2)*L)*2^(1/2)*omega^(1/2)/alpha^(1/2)&
```

```
-(1/2+1/2*I)*C_4*sin((1/2+1/2*I)*2^(1/2)*omega^(1/2)/alpha^(1/2)*L)*2^(1/2)*omega^(1/2)/alpha^(1/2)) =&
h_bar_o*(C_3*sin((1/2+1/2*I)*2^(1/2)*omega^(1/2)/alpha^(1/2)*L)+&
C_4*cos((1/2+1/2*I)*2^(1/2)*omega^(1/2)/alpha^(1/2)*L))

"boundary condition #2 from Maple"
h_bar_c*(DT_c*I-C_4) = (-1/2-1/2*I)*k*C_3*2^(1/2)*omega^(1/2)/alpha^(1/2)
```

The length of these expressions should provide some appreciation for the amount of algebra that can be avoided by using Maple and EES. Solving these expressions for C_3 and C_4 shows that these constants are complex values, C_3 = (-46.19-38.9 i) K and C_4 = (-38.9+46.19 i) K. The solution for B can be copied from Maple and pasted into EES and evaluated at a particular position:

```
x_bar=1                                                          "dimensionless position"
x=x_bar*L                                                        "position"
"solution for B, from Maple"
B = C_3*sin((1/2+1/2*I)*2^(1/2)*omega^(1/2)/alpha^(1/2)*x)+&
          C_4*cos((1/2+1/2*I)*2^(1/2)*omega^(1/2)/alpha^(1/2)*x)
```

The complex conjugate solution, Eq. (3-483), can be evaluated at a particular position and time (note that time is normalized by the cycle time):

```
time_bar=0.9                                                     "dimensionless time"
time=time_bar*2*pi/omega                                         "time"
T_h_cc=B*exp(i*omega*time)                                       "complex conjugate solution"
```

which leads to a complex solution for $T_{h,cc}$. According to Eq. (3-459), the real portion of the complex conjugate solution is the sustained homogeneous portion of the solution, T_h. It would be difficult to determine a symbolic expression for the complex conjugate solution, much less carry out the additional complex algebra that would be required to determine its real part symbolically. However, by evaluating the complex conjugate solution in EES, it is possible to use the real command to determine the real portion of the solution:

```
T_h=real(T_h_cc)                  "real portion of the complex conjugate solution"
```

The sustained solution is constructed by adding the particular solution, X from Eq. (3-463), to the homogeneous solution, T_h, according to Eq. (3-459):

```
Xp=C_1*x+C_2                                                     "particular portion of the solution"
T=Xp+T_h                                                         "sustained solution"
```

A parametric table is generated that contains the variables x_bar_r, x_bar_i, and T_r; the value of x_bar_r is varied from 0 to 1 and the value of x_bar_i is set to zero. The temperature as a function of dimensionless position for various dimensionless times is shown in Figure 3-39.



Figure 3-39: Temperature as a function of dimensionless position for various values of dimensionless time.

A second parametric table is generated that contains the variables t_bar_r, t_bar_i, and T_r; the value of t_bar_r is varied from 0 to 1 and the value of t_bar_i is set to zero. The temperature as a function of dimensionless time for various dimensionless positions is shown in Figure 3-40.



Figure 3-40: Temperature as a function of dimensionless time for various values of dimensionless position.

Figure 3-39 and Figure 3-40 should be examined to ensure that they make sense relative to our conceptual understanding of transient conduction. The thermal penetration depth associated with the time required for half of a cycle is:

$$\delta_t \approx 2\sqrt{\frac{\pi\,\alpha}{\omega}} \tag{3-491}$$

and so the dimensionless depth at which the influence of the cylinder temperature variation should no longer be felt is δ_t/L.

```
delta_t=2*sqrt(2*pi*alpha/omega)          "approximate extent of thermal wave"
delta_toverL=delta_t/L                    "approximate dimensionless extent of thermal wave"
```

which leads to δ_t/L = 0.27. Figure 3-39 shows the impact of the time varying cylinder temperature has decayed to near zero at about x/L = 0.30.

An additional sanity check can be accomplished by calculating the thermal resistances that characterize this problem. The resistances (per unit cylinder area) to convection to the cylinder gas and the cooling fluid are:

$$R_{conv,c} = \frac{1}{\overline{h}_c} \tag{3-492}$$

$$R_{conv,o} = \frac{1}{\overline{h}_o} \tag{3-493}$$

The resistance to conduction through the cylinder is:

$$R_{cond,L} = \frac{L}{k} \tag{3-494}$$

The resistance to conduction to the thermally affected region (i.e., the material at $x < \delta_t$) is:

$$R_{cond,\delta_t} = \frac{\delta_t}{k}$$

```
R_conv_c=1/h_bar_c          "resistance to convection to the cylinder"
R_cond_dt=delta_t/k         "resistance to conduction to the thermally affected region"
R_cond_L=L/k                "resistance to conduction across the cylinder"
R_conv_o=1/h_bar_o          "resistance to convection to the oil cooled side"
```

which leads to $R_{conv,c}$ = 0.00017 K-m^2/W, $R_{conv,o}$= 0.0029 K-m^2/W, $R_{cond,L}$ = 0.00030 K-m^2/W, and R_{cond,δ_t} = 0.000081 K-m^2/W. Note from Figure 3-39 that the largest temperature drop, on average, is related to convection from the lower surface of the piston; this is consistent with the fact that $R_{conv,o}$ is larger than $R_{conv,c}$ or $R_{cond,L}$. Also, note that the amplitude of the cylinder gas oscillation is much larger than the amplitude of the wall temperature variation (see Figure 3-40). This is consistent with the fact that $R_{conv,c}$ is much larger than R_{cond,δ_t}. The solution matches our intuition and makes physical sense.